version 14.2
clear
set more off

*     ****************************************************************  *
*     ****************************************************************  *
*       Filename:       pfp_9_analysis_WGI_postcon7+10.do               *
*       Date:           06/03/2017                                      *
*       Author:         Felix Haass, Martin Ottmann                     *
*     ****************************************************************  *
*     ****************************************************************  *



// Alternative Model Specifications
// Finally, we explore whether our estimation results continue to hold if we move 
// beyond the common practice in power-sharing research of only looking at the 
// first five years of a post-conflict period. We therefore estimate our models 
// using a sample with seven post-conflict years and ten post-conflict years. 
// We also include a variable measuring the number of years since executive 
// power-sharing has ended to account for time dependency.



******* OVERVIEW *******
******* (1) load data *******
******* (2) Appendix Table 8: OLS regressions 7 and 10 post-conflict years *******
******* (3) Appendix Figure 7a: interaction b/w cabinetINC and WBnatres_ln *******
******* (4) Appendix Figure 7b: interaction b/w cabinetINC and aid_gdp_ln *******





******* (1) load data *******

use profits_from_peace_replication_dataset.dta, clear



******* (2) Table 8: OLS regressions 7 and 10 post-conflict years *******

local model1 "cabinetINC WBnatres_ln aid_gdp_ln WBDI_gdppc_ln WBDI_poptotal_ln ethnic polity2 postcabtime"
local model2 "cabinetINC WBnatres_ln aid_gdp_ln WBDI_gdppc_ln WBDI_poptotal_ln ethnic polity2 cabXnatres postcabtime"
local model3 "cabinetINC WBnatres_ln aid_gdp_ln WBDI_gdppc_ln WBDI_poptotal_ln ethnic polity2 cabXaid postcabtime"
 * sample: postconflict7
forval i = 1/3 {
   quietly reg WGI_corrupt `model`i'' if pcy < 8, robust cluster(gwno)
   estimates store postcon7`i'
 }
* sample: postconflict10
forval i = 1/3 {
   quietly reg WGI_corrupt `model`i'', robust cluster(gwno)
   estimates store postcon10`i'
 }
estout ///
postcon71 postcon72 postcon73 ///
postcon101 postcon102 postcon103 ///
, cells(b(star fmt(2)) se(par fmt(2))) starlevels(* 0.1 ** 0.05 *** 0.01) stats(r2 r2_a N, fmt(2 2 0)) legend
esttab ///
postcon71 postcon72 postcon73 postcon101 postcon102 postcon103 ///
using appendix_table_8.tex ///
, b(2) se(2) star(* 0.1 ** 0.05 *** 0.01) ///
stats(N r2_a, fmt(0 2) layout("\multicolumn{1}{c}{@}" "\multicolumn{1}{c}{@}") labels(`"Observations"' `"Adjusted R\textsuperscript{2}"')) ///
title(OLS Regression on a Sample Including the First Seven and Ten Post-Conflict Years) ///
mtitles("\specialcell{Model 1\\(7 Years)}" "\specialcell{Model 2\\(7 Years)}" "\specialcell{Model 3\\(7 Years)}" "\specialcell{Model 1\\(10 Years)}" "\specialcell{Model 2\\(10 Years)}" "\specialcell{Model 3\\(10 Years)}") ///
order(cabinetINC WBnatres_ln cabXnatres aid_gdp_ln cabXaid WBDI_gdppc_ln polity2 ethnic WBDI_poptotal_ln) ///
label nogaps nonumbers booktabs fragment ///
replace



******* (3) Figure 7a: interaction b/w cabinetINC and WBnatres_ln *******

quietly ///
reg WGI_corrupt ///
i.cabinetINC##c.WBnatres_ln ///
c.polity2 c.WBDI_gdppc_ln c.WBDI_poptotal_ln c.ethnic c.postcabtime ///
c.aid_gdp_ln ///
if pcy < 8, robust cluster(gwno)
quietly margins, dydx(cabinetINC) at(WBnatres_ln=(0(0.1)4.5) (mean)_c) vsquish
matrix b = r(b)'
matrix list b
matrix b = b[47...,1]
matrix list b
matrix at = r(at)
matrix list at
matrix at = at[1...,"WBnatres_ln"]
matrix list at
matrix v = r(V)
matrix list v
matrix v = v[47...,47...]
matrix list v
matrix se = vecdiag(cholesky(diag(vecdiag(v))))'
matrix list se
matrix d = at,b,se
matrix list d
svmat d, names(d)
generate ul = d2 + 1.96 * d3
generate ll = d2 - 1.96 * d3
clist d1-ll in 1/12
gen yline = 0
graph twoway ///
   (hist WBnatres_ln if pcy < 6, width(0.1) percent color(gs14) yaxis(2) ytitle("Percentage of Observations", axis(2))) ///
   (line d2 d1, clpattern(solid) clwidth(medium) clcolor(black) yaxis(1)) ///
   (line ul d1, clpattern(shortdash) clwidth(thin) clcolor(black)) ///
   (line ll d1, clpattern(shortdash) clwidth(thin) clcolor(black)) ///
   (line yline d1, clwidth(thin) clcolor(black) clpattern(solid)), ///
   xtitle("Natural Resource Wealth (ln)") ///
   ytitle("Conditional Marginal Effect of Executive PS", axis(1)) ///
   title("Seven Post-Conflict Years", size(medium)) ///
   ylabel(-1(.5).5) ///
   xlabel(0(1)4.5) ///
   xsize(8) ysize(6) ///
   scheme(lean1) legend(off)
graph save WGI_cabXnatres_postconflict7.gph, replace
drop d1 d2 d3 ul ll yline

quietly ///
reg WGI_corrupt ///
i.cabinetINC##c.WBnatres_ln ///
c.polity2 c.WBDI_gdppc_ln c.WBDI_poptotal_ln c.ethnic c.postcabtime ///
c.aid_gdp_ln ///
, robust cluster(gwno)
quietly margins, dydx(cabinetINC) at(WBnatres_ln=(0(0.1)4.5) (mean)_c) vsquish
matrix b = r(b)'
matrix list b
matrix b = b[47...,1]
matrix list b
matrix at = r(at)
matrix list at
matrix at = at[1...,"WBnatres_ln"]
matrix list at
matrix v = r(V)
matrix list v
matrix v = v[47...,47...]
matrix list v
matrix se = vecdiag(cholesky(diag(vecdiag(v))))'
matrix list se
matrix d = at,b,se
matrix list d
svmat d, names(d)
generate ul = d2 + 1.96 * d3
generate ll = d2 - 1.96 * d3
clist d1-ll in 1/12
gen yline = 0
graph twoway ///
   (hist WBnatres_ln if pcy < 6, width(0.1) percent color(gs14) yaxis(2) ytitle("Percentage of Observations", axis(2))) ///
   (line d2 d1, clpattern(solid) clwidth(medium) clcolor(black) yaxis(1)) ///
   (line ul d1, clpattern(shortdash) clwidth(thin) clcolor(black)) ///
   (line ll d1, clpattern(shortdash) clwidth(thin) clcolor(black)) ///
   (line yline d1, clwidth(thin) clcolor(black) clpattern(solid)), ///
   xtitle("Natural Resource Wealth (ln)") ///
   ytitle("Conditional Marginal Effect of Executive PS", axis(1)) ///
   title("Ten Post-Conflict Years", size(medium)) ///
   ylabel(-1(.5).5) ///
   xlabel(0(1)4.5) ///
   xsize(8) ysize(6) ///
   scheme(lean1) legend(off)
graph save WGI_cabXnatres_postconflict10.gph, replace
drop d1 d2 d3 ul ll yline

graph combine ///
WGI_cabXnatres_postconflict7.gph ///
WGI_cabXnatres_postconflict10.gph, ///
ycommon xcommon rows(1) iscale(*0.9) ///
title(, fcolor(white) lcolor(white)) xsize(14) ysize(6) graphregion(margin(zero) ///
fcolor(white) lcolor(white) ifcolor(white) ilcolor(white)) plotregion(fcolor(white) lcolor(white) ifcolor(white) ilcolor(white))
graph export appendix_figure_7a.pdf, replace
erase WGI_cabXnatres_postconflict7.gph
erase WGI_cabXnatres_postconflict10.gph



******* (4) Appendix Figure 7b: interaction b/w cabinetINC and aid_gdp_ln *******

quietly ///
reg WGI_corrupt ///
i.cabinetINC##c.aid_gdp_ln ///
c.polity2 c.WBDI_gdppc_ln c.WBDI_poptotal_ln c.ethnic c.postcabtime ///
c.WBnatres_ln ///
if pcy < 8, robust cluster(gwno)
quietly margins, dydx(cabinetINC) at(aid_gdp_ln=(-11(0.2)1) (mean)_c) vsquish
matrix b = r(b)'
matrix list b
matrix b = b[62...,1]
matrix list b
matrix at = r(at)
matrix list at
matrix at = at[1...,"aid_gdp_ln"]
matrix list at
matrix v = r(V)
matrix list v
matrix v = v[62...,62...]
matrix list v
matrix se = vecdiag(cholesky(diag(vecdiag(v))))'
matrix list se
matrix d = at,b,se
matrix list d
svmat d, names(d)
generate ul = d2 + 1.96 * d3
generate ll = d2 - 1.96 * d3
clist d1-ll in 1/12
gen yline = 0
graph twoway ///
   (hist aid_gdp_ln if pcy < 6, width(0.3) percent color(gs14) yaxis(2) ytitle("Percentage of Observations", axis(2))) ///
   (line d2 d1, clpattern(solid) clwidth(medium) clcolor(black) yaxis(1)) ///
   (line ul d1, clpattern(shortdash) clwidth(thin) clcolor(black)) ///
   (line ll d1, clpattern(shortdash) clwidth(thin) clcolor(black)) ///
   (line yline d1, clwidth(thin) clcolor(black) clpattern(solid)), ///
   xtitle("Foreign Aid Income (ln)") ///
   ytitle("Conditional Marginal Effect of Executive PS", axis(1)) ///
   title("Seven Post-Conflict Years", size(medium)) ///
   ylabel(-3(.5)1) ///
   xlabel(-11(2)1) ///
   xsize(8) ysize(6) ///
   scheme(lean1) legend(off)
graph save WGI_cabXaid_postconflict7.gph, replace
drop d1 d2 d3 ul ll yline

quietly ///
reg WGI_corrupt ///
i.cabinetINC##c.aid_gdp_ln ///
c.polity2 c.WBDI_gdppc_ln c.WBDI_poptotal_ln c.ethnic c.postcabtime ///
c.WBnatres_ln ///
, robust cluster(gwno)
quietly margins, dydx(cabinetINC) at(aid_gdp_ln=(-11(0.2)1) (mean)_c) vsquish
matrix b = r(b)'
matrix list b
matrix b = b[62...,1]
matrix list b
matrix at = r(at)
matrix list at
matrix at = at[1...,"aid_gdp_ln"]
matrix list at
matrix v = r(V)
matrix list v
matrix v = v[62...,62...]
matrix list v
matrix se = vecdiag(cholesky(diag(vecdiag(v))))'
matrix list se
matrix d = at,b,se
matrix list d
svmat d, names(d)
generate ul = d2 + 1.96 * d3
generate ll = d2 - 1.96 * d3
clist d1-ll in 1/12
gen yline = 0
graph twoway ///
   (hist aid_gdp_ln if pcy < 6, width(0.3) percent color(gs14) yaxis(2) ytitle("Percentage of Observations", axis(2))) ///
   (line d2 d1, clpattern(solid) clwidth(medium) clcolor(black) yaxis(1)) ///
   (line ul d1, clpattern(shortdash) clwidth(thin) clcolor(black)) ///
   (line ll d1, clpattern(shortdash) clwidth(thin) clcolor(black)) ///
   (line yline d1, clwidth(thin) clcolor(black) clpattern(solid)), ///
   xtitle("Foreign Aid Income (ln)") ///
   ytitle("Conditional Marginal Effect of Executive PS", axis(1)) ///
   title("Ten Post-Conflict Years", size(medium)) ///
   ylabel(-3(.5)1) ///
   xlabel(-11(2)1) ///
   xsize(8) ysize(6) ///
   scheme(lean1) legend(off)
graph save WGI_cabXaid_postconflict10.gph, replace
drop d1 d2 d3 ul ll yline

graph combine ///
WGI_cabXaid_postconflict7.gph ///
WGI_cabXaid_postconflict10.gph, ///
ycommon xcommon rows(1) iscale(*0.9) ///
title(, fcolor(white) lcolor(white)) xsize(14) ysize(6) graphregion(margin(zero) ///
fcolor(white) lcolor(white) ifcolor(white) ilcolor(white)) plotregion(fcolor(white) lcolor(white) ifcolor(white) ilcolor(white))
graph export appendix_figure_7b.pdf, replace
erase WGI_cabXaid_postconflict7.gph
erase WGI_cabXaid_postconflict10.gph
